Soluzione di un sistema 4x4 con Gauss

Enrico Bertolazzi
Esercitazioni Maple di Geometria

```
> with(LinearAlgebra) :
> # definisce il termine noto
  b := <1,2,2,1> ;
```

$$b := \begin{bmatrix} 1 \\ 2 \\ 2 \\ 1 \end{bmatrix} \quad (1)$$

```
> # matrice dei coefficienti + termine noto
  A := <<1,2,3,4>|<1,2,0,5>|<0,0,-3,-4>|<1,2,0,8>|b>;
```

$$A := \begin{bmatrix} 1 & 1 & 0 & 1 & 1 \\ 2 & 2 & 0 & 2 & 2 \\ 3 & 0 & -3 & 0 & 2 \\ 4 & 5 & -4 & 8 & 1 \end{bmatrix} \quad (2)$$

```
> # prima matrice per azzerare gli elementi [2,1] [3,1] [4,1]
  L1 := <<1,-2,-3,-4>|<0,1,0,0>|<0,0,1,0>|<0,0,0,1>> ;
```

$$L1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ -2 & 1 & 0 & 0 \\ -3 & 0 & 1 & 0 \\ -4 & 0 & 0 & 1 \end{bmatrix} \quad (3)$$

```
> # applica la trasformazione
  L1A := L1.A ;
```

$$L1A := \begin{bmatrix} 1 & 1 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & -3 & -3 & -3 & -1 \\ 0 & 1 & -4 & 4 & -3 \end{bmatrix} \quad (4)$$

```
> # matrice per scambiare la riga 2 con la riga 4
  P1 := <<1,0,0,0>|<0,0,0,1>|<0,0,1,0>|<0,1,0,0>> ;
```

(5)

$$P1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix} \quad \textbf{(5)}$$

```
> # esegue lo scambio
  P1L1A := P1.L1A ;
```

$$P1L1A := \begin{bmatrix} 1 & 1 & 0 & 1 & 1 \\ 0 & 1 & -4 & 4 & -3 \\ 0 & -3 & -3 & -3 & -1 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad \textbf{(6)}$$

```
> # seconda matrice per azzerare gli elementi [1,2] [3,2] [4,2]
  L2 := <<1,0,0,0>|<-1,1,3,0>|<0,0,1,0>|<0,0,0,1>> ;
```

$$L2 := \begin{bmatrix} 1 & -1 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 3 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(7)}$$

```
> # applica la trasformazione
  L2P1L1A := L2.P1L1A ;
```

$$L2P1L1A := \begin{bmatrix} 1 & 0 & 4 & -3 & 4 \\ 0 & 1 & -4 & 4 & -3 \\ 0 & 0 & -15 & 9 & -10 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad \textbf{(8)}$$

```
> # pone a 1 l'elemento [3,3]
  D1        := DiagonalMatrix(<1,1,-1/15,1>);
  D1L2P1L1A := D1.L2P1L1A ;
```

$$D1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -\frac{1}{15} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(9)}$$

$$D1L2P1L1A := \begin{bmatrix} 1 & 0 & 4 & -3 & 4 \\ 0 & 1 & -4 & 4 & -3 \\ 0 & 0 & 1 & -\frac{3}{5} & \frac{2}{3} \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> # terza matrice per azzerare gli elementi [1,3] [2,3] [4,3]
  L3 := <<1,0,0,0>|<0,1,0,0>|<-4,4,1,0>|<0,0,0,1>> ;
```

$$L3 := \begin{bmatrix} 1 & 0 & -4 & 0 \\ 0 & 1 & 4 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \tag{10}$$

```
> # applica la trasformazione
  L3D1L2P1L1A := L3.D1L2P1L1A ;
```

$$L3D1L2P1L1A := \begin{bmatrix} 1 & 0 & 0 & -\frac{3}{5} & \frac{4}{3} \\ 0 & 1 & 0 & \frac{8}{5} & -\frac{1}{3} \\ 0 & 0 & 1 & -\frac{3}{5} & \frac{2}{3} \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \tag{11}$$

```
> # il sistema ora e` immediatamente risolubile
  RES := L3D1L2P1L1A.<x[1],x[2],x[3],x[4],-1> ;
```

$$RES := \begin{bmatrix} x_1 - \frac{4}{3} - \frac{3}{5}\, x_4 \\ \frac{1}{3} + x_2 + \frac{8}{5}\, x_4 \\ -\frac{2}{3} + x_3 - \frac{3}{5}\, x_4 \\ 0 \end{bmatrix} \tag{12}$$

```
> SOL1 := isolate(RES[1],x[1]) ;
  SOL2 := isolate(RES[2],x[2]) ;
  SOL3 := isolate(RES[3],x[3]) ;
```

$$SOL1 := x_1 = \frac{4}{3} + \frac{3}{5}\, x_4 \tag{13}$$

$$SOL2 := x_2 = -\frac{1}{3} - \frac{8}{5}\, x_4$$

$$SOL3 := x_3 = \frac{2}{3} + \frac{3}{5}\, x_4$$

```
> # scrivo la soluzione come somma
  # soluzione particolare + kernel

  MYSOL := <rhs(SOL1),rhs(SOL2),rhs(SOL3),x[4]> ;
```

$$MYSOL := \begin{bmatrix} \frac{4}{3} + \frac{3}{5} x_4 \\ -\frac{1}{3} - \frac{8}{5} x_4 \\ \frac{2}{3} + \frac{3}{5} x_4 \\ x_4 \end{bmatrix} \tag{14}$$

```
> # calcolo la soluzione conle primitive di MAPLE
  MAPLESOL := LinearSolve(A,free='x') ;
```

$$MAPLESOL := \begin{bmatrix} x_1 \\ \frac{29}{9} - \frac{8}{3} x_1 \\ x_1 - \frac{2}{3} \\ \frac{5}{3} x_1 - \frac{20}{9} \end{bmatrix} \tag{15}$$

```
> # sono apparentemente diverse, ma ponendo
  TRAS := x[4] = MAPLESOL[4] ;
```

$$TRAS := x_4 = \frac{5}{3} x_1 - \frac{20}{9} \tag{16}$$

```
> # e sostituendo otteniamo
  subs(TRAS,MYSOL) ;
```

$$\begin{bmatrix} x_1 \\ \frac{29}{9} - \frac{8}{3} x_1 \\ x_1 - \frac{2}{3} \\ \frac{5}{3} x_1 - \frac{20}{9} \end{bmatrix} \tag{17}$$

```
> # cioe` le soluzioni coincidono
```